

```

1 *****
2 *
3 *      WPSave20 - An AppleWorks Init
4 *
5 *      - When working in the OA-N 'Change Filename'
6 *      command dialog in the Word Processor
7 *      REVIEW/ADD/CHANGE Window, adds the
8 *      ability to specify the both the
9 *      type of saved file (AWP vs. Text) and,
10 *      in the case of Text files, the type of
11 *      line ending characters, as follows:
12 *
13 *      1. AWP (Word Processor File)
14 *      2. TXT (Apple Text File - CR)
15 *      3. DOS (MS-DOS/Windows Text File - CR/LF)
16 *      4. NIX (Unix Text File - LF)
17 *      5. MLN (Merlin Assembler Source File)
18 *         (high bits set / <TAB> = $A0)
19 *
20 *      - After providing the opportunity to change
21 *      the name of the file via the <OA-N> key
22 *      sequence, pressing <OA-Return> presents
23 *      a horizontal menu allowing the selection
24 *      of the desired save type.
25 *
26 *      - Type of file is displayed after the
27 *      REVIEW/ADD/CHANGE in top line of Display
28 *      (provided WPStatus30 Init is installed)
29 *
30 *      Version 2.0 (for AppleWorks Version 5.1)
31 *      (c) 2017 Hugh Hood
32 *
33 *      - Seg $17 is patch-in segment.
34 *
35 *****

```

```

36
37      TR          ADR          ; truncate bank address
38
39      XC          ; enable 65C02 code
40
41 * Equates *
42
43 FoundRtn      EQU          $87          ; non-zero if <Rtn> pressed
44                                     ; (but, does NOT set for <OA-Rtn>
45 FoundEsc      EQU          $88          ; non-zero if <ESC> pressed
46 CArg          EQU          $9E          ; & $9F - pointer
47                                     ; (use to dereference pointer)
48 LastChar      EQU          $360        ; last character entered from keyboard
49                                     ; (high bit set for 'OA' characters
50 DTFStatus     EQU          $C6C        ; current file status flag byte:
51                                     ; ($01/0001) or $03/0011) - new
52                                     ; ($02/0010) or $06/0110) - changed
53                                     ; ($04/0100) - saved
54                                     ; ($00/0000) - unchanged
55 NewStr        EQU          $D85        ; used by WP to contain 'Line/Column' String
56 OldStr        EQU          $E05        ; used by WP to contain 'Line/Column' String

```



```

113                                     ;
114         ORG             InitAdr       ; ($4000)
115         TYP             $06           ; create binary file
116
117 *****
118 *             Init Header             *
119 *****
120 START
121         JMP             IStart        ; skip over header
122
123 **-----
124
125         ASC             'mb'          ; Init ID Bytes (AW 5.1)
126         DB              $14          ; Init Version - programmer assigned
127                                     ; e.g. - $0A/1.0 $0C/1.2 $19/2.5
128         STR             'WPSetSave20' ; Init Screen Name
129         HEX             0000         ; Header End Bytes
130
131 **-----
132
133 IStart
134
135         LDA             AVersion      ; AppleWorks version #
136         CMP             #$33         ; Is it Version 5.1?
137         BNE            Done          ; disregard - wrong version
138
139 PatchH17      JSR       imSavePatch  ; call patch manager
140              DW        Code1        ; beginning of patch1 code ($40xx)
141              DW        Patch1End-PatchAdr
142              DW        $0017        ; length of patch code
143              DW        $0017        ; SEG number to patch
144                                     ; ($17 = AWP Util SEG)
145
146 PatchH15      JSR       imSavePatch  ; call patch manager
147              DW        Code1+Patch1End-PatchAdr
148              DW        MoveStart-PatchAdr+Patch2End-Patch2Run
149              DW        $0017        ; length of patch code
150              DW        $0017        ; SEG number to patch
151                                     ; ($17 = AWP Util SEG)
152 Done          RTS
153
154 **-----
155
156 Code1         EQU       *            ; (will be $40xx)
157
158         ORG             PatchAdr     ; (Patching Code is moved and run
159                                     ; @ $BB00 by Init Manager)
160
161 HookBytes     HEX      0000         ; first bytes for $17 Patch
162              LDA      #$4C          ; JMP instruction
163              STA      PatchPoint    ; $8FD9 in SEG $17
164              LDA      #Patch2Run     ; low byte of new code
165              STA      PatchPoint+1

```

```

166          LDA          #>Patch2Run      ; high byte of new code
167          STA          PatchPoint+2
168
169          LDA          #$4C              ; JMP instruction
170          STA          PatchPoint2      ; $8FCE in SEG $17
171          LDA          #CmdHandler       ; low byte of new code
172          STA          PatchPoint2+1
173          LDA          #>CmdHandler     ; high byte of new code
174          STA          PatchPoint2+2
175
176          RTS
177
178 Patch1End  EQU          *
179
180 **-----
181
182          ORG          PatchAdr          ; (Patching Code is moved and run
183                                     ; @ $BB00 by Init Manager)
184
185 MoveCall  HEX          0000            ; first bytes for $17 Patch
186          JSR          MvLeftRtn        ; move new code to run location
187          DA          #Patch2Run        ; ($9D00)
188          DA          #MoveStart        ;
189          DA          Patch2End-Patch2Run
190
191          RTS                          ; patch hook-in done
192
193 **-----
194
195 MoveStart EQU          *              ; (will be $BBxx)
196
197          ORG          Patch2Run        ; ($9D00)
198
199          LDA          LastChar         ; check last keypress
200          CMP          #$8D            ; was it <0A-Rtn>?
201          BEQ          PatchStart      ; do 'Change Save Type'
202
203 SkipType  JMP          PopStack        ; skip 'Change Save Type'
204
205 PatchStart JSR         PopStack        ; remove last 'title' ('Change Filename')
206
207          LDA          MMInDoc          ; get file flag byte
208          CMP          #$20            ; is it Apple Text (CR)?
209          BEQ          TXTTyp          ;
210          CMP          #$24            ; is it Merlin Source?
211          BEQ          MLNTyp         ;
212          CMP          #$28            ; is it Unix Text (LF)?
213          BEQ          NIXTyp         ;
214          CMP          #$30            ; is is MS-DOS Text (CR/LF)?
215          BEQ          DOSTyp         ;
216
217 AWPTyp   LDA          #1              ; menu item #1
218          STA          Default+1       ; default for menu routine
219          BRA          :B
220
221 TXTTyp   LDA          #2              ; menu item #2

```

```

222          STA          Default+1      ; default for menu routine
223          BRA          :B
224
225 MLNTyp     LDA          #5           ; menu item #5
226          STA          Default+1      ; default for menu routine
227          BRA          :B
228
229 NIXTyp     LDA          #4           ; menu item #4
230          STA          Default+1      ; default for menu routine
231          BRA          :B
232
233 DOSTyp     LDA          #3           ; menu item #3
234          STA          Default+1      ; default for menu routine
235
236
237 :B         JSR          newPushStack ; add level to Escape road map
238          DA          TitleStr
239          JSR          WriteCom
240          DA          PromptStr
241 Default    LDA          #1           ; default selection # (changes)
242          JSR          GetMenuBar2
243          DA          GoBack
244          DB          5               ; # of selections
245          DA          MenuAWPStr
246          DA          SetType
247          DA          MenuTXTStr
248          DA          SetType
249          DA          MenuDOSStr
250          DA          SetType
251          DA          MenuNIXStr
252          DA          SetType
253          DA          MenuMLNStr
254          DA          SetType
255
256
257
258 SetType    LDA          GetMenuBar+1 ; low byte of pointer to actual routine
259          STA          CArg           ; low byte pointer for indirect reference
260          LDA          GetMenuBar+2 ; high byte of pointer to actual routine
261          STA          CArg+1        ; high byte pointer for indirect reference
262          DEC          CArg           ; subtract first of (2) bytes for location
263          ; of stored menu selection
264          LDA          CArg           ; load for CMP operation
265          CMP          #$FF          ; did we cross page boundary?
266          BNE          :C           ; if not, do one more decrement
267          DEC          CArg+1        ; cross page - subtract (1) from high byte
268 :C         DEC          CArg         ; subtract second of (2) bytes
269          CMP          #$FF          ; did we cross page boundary?
270          BNE          :D           ; if not, proceed with rest
271          DEC          CArg+1        ; cross page - subtract (1) from high byte
272 :D         LDA          (CArg)       ; retrieve selection #
273          TAX          ; put it in 'X' for CPX operations
274          LDA          MMInDoc       ; read AWP file flag byte
275          CPX          Default+1    ; was selection the initial default?
276          BEQ          GoBack        ; if so, then done
277          JSR          MadeAChange   ; flag change (DTFStatus)

```

```

278 CPX #2 ; was selection TXT?
279 BEQ SetTXT ; set TXT flag
280 CPX #3 ; was selection MS-DOS?
281 BEQ SetDOS ; set DOS flag
282 CPX #4 ; was selection UNIX?
283 BEQ SetNIX ; set NIX flag
284 CPX #5 ; was selection Merlin Source?
285 BEQ SetMLN ; set MLN flag
286
287 ; if not TXT/DOS/NIX/MLN, then is AWP
288 SetAWP AND #%00000001 ; force all bits low except Bit 0,
289 ; in case this is a Mail Merge AWP file
290 STA MMInDoc ;
291 BRA GoBack
292
293 SetTXT LDA #%00100000
294 STA MMInDoc
295 BRA GoBack
296
297 SetDOS LDA #%00110000
298 STA MMInDoc
299 BRA GoBack
300
301 SetNIX LDA #%00101000
302 STA MMInDoc
303 BRA GoBack
304
305 SetMLN LDA #%00100100
306 STA MMInDoc
307
308
309 **-----
310
311 GoBack JMP PopStack ; go back to original code
312
313 **-----
314
315 CmdHandler EQU *
316
317 LDA LastChar ; check last keypress
318 CMP #$8D ; was it <OA-Rtn>?
319 BNE :NextCheck1
320 JMP PatchStart ; proceed to 'Change Save Type'
321
322 :NextCheck1 LDA FoundEsc ; was it <ESC> key?
323 ORA FoundRtn ; check <Rtn> key as well
324 BNE :NextCheck2
325
326 JMP ErrorHandler ; back to 'Change filename' routine
327
328 :NextCheck2 JMP SkipType ; back to AppleWorks
329
330
331 **-----
332
333 PromptStr STR 'Save Type' ; prompt for horizontal menu

```

```
334 MenuAWPStr    STR      'Word Processor'  
335 MenuTXTStr   STR      'Apple Text'  
336 MenuDOSStr   STR      'MS-DOS Text'  
337 MenuNIXStr   STR      'Unix Text'  
338 MenuMLNStr   STR      'Source-Merlin'  
339 TitleStr     STR      'Change Save Type'
```

```
340  
341 **-----
```

```
342  
343 Patch2End     EQU      *                ;  
344              SAV      I.WPSETSAVE20  
345              LST      OFF  
346  
347              END  
348  
349
```